
Un corrigé de TP en SCILAB sur le chapitre 2

Temps de calcul pour la décomposition LU et Cholesky (exercice 3)

```
//comparaison des temps de calcul pour la decomposition LU et de Cholesky
clear
Tabiter=[];
tempslu=[];
tempchol=[];
tempstheo=[];
//stacksize(20000000) //pour augmenter la memoire
kmax=12;//par default, le kmax maximum possible est 9 et non 12 a cause de la memoire
//utiliser la ligne stacksize ci-dessus pour l'augmenter
//sur cette machine, je n'ai pas reussi a aller jusqu'a kmax=13
for k=2:kmax
    n=2^k;
    A=diag(2*ones(n,1))-diag(ones(n-1,1),-1)-diag(ones(n-1,1),1);
    tic
    [L,U]=lu(A);
    tlu=toc();
    Tabiter=[Tabiter n];
    tempslu=[tempslu tlu];
    tic
    R=chol(A);
    tchol=toc();
    tempchol=[tempchol tchol];
    tempstheo=[tempstheo n^3/10^8];
end
max(tempslu)
rapport=tempslu(6:kmax-1)./tempchol(6:kmax-1)
clf
plot2d(Tabiter',[tempslu' tempchol' tempstheo'] ,logflag='ll')
legends(['LU'; 'Cholesky'; 'O(n3)'],[1 2 3],opt='ul')
xtitle('Decomposition LU','taille n','temps cpu')
```

Ce programme produit la figure suivante (au moins 2 minutes d'exécution sur mon ordinateur ; pour avoir une première exécution plus rapide et sans problèmes de mémoire, fixer $kmax$ à 9)

On constate que les temps de calcul sont bien en $O(n^3)$, comme prévu par la théorie.

Le rapport entre le temps de calcul pour la decomposition LU et celle de Cholesky est proche de 2, ce qui confirme expérimentalement la théorie.




Figure 1: Temps de calcul pour les decomposition LU et Cholesky